問合せ言語の意味論 3 — 層状でない場合の扱い

層状でない例

ニムゲーム



- ２人のプレーヤーが交互に局面を動かす手をうつ.
- 動かせない局面は負け.
- 負けの局面に動かす手がある局面は勝ち.
- 勝ちの局面だけにしか動かす手がない局面は負け.

- a, d は負けの局面.
- b, c は勝ちの局面.
- win(X) :- move(X,Y) & NOT win(Y)
  move(a, b)
  move(a, c)
  move(b, c)
  move(b, d)
  move(c, d)

ニムゲーム続き



- c は負けの局面
- b は勝ちの局面,
- a,d,e は千日手の局面.
- win(X) :- move(X,Y) & NOT win(Y)
  move(a,b)
  move(a,d)
  move(b,c)
  move(d,e)
  move(e,d)
  move(e,b)

局所的層状 (locally stratified) 論理プログラム

- 任意のルールについて、すべての基礎代入例を生成.
- プログラムに出現しない EDB ファクトは偽と考える.
  本体に偽となる EDB ファクトを含む基礎代入例を除く.
- 基礎原子式を述語とみなして依存グラフをつくる.
  “-” のラベルのついた有向辺を含む閉路が存在しない.

例



- win(X) :- move(X,Y) & NOT win(Y)
  EDB = { move(a,b), move(b,c), move(a,c) }
- r_1 : win(a) :- move(a,b) & NOT win(b)
  r_2 : win(b) :- move(b,c) & NOT win(c)
  r_3 : win(a) :- move(a,c) & NOT win(c)
- 例えば, win(b) :- move(b,a) & NOT win(a) は move(b,a) が偽なので生成しない.
- win(a) 層 2, win(b) 層 1, win(c) 層 0

局所的層状エルブランモデル

層状エルブランモデルと同様に, 下の層からボトムアップに計算して得られるモデル.

1. win(c) は偽. win(c) を生成するルールがないから.
2. ルール r_2 より win(b) は真.
3. ルール r_3 より win(a) は真.

１ステップ推論の不動点としての定式化

- 局所的層状の概念は, すべての基礎代入例を生成した後に, 層が定義できるか否かを考える.
- 層という概念を使わないで意味論を構築できないか？

Stable Model

基礎原子式の集合 M にルールを１ステップ適用すると, ちょうど M が推論できるとき M は Stable Model であるという.

例

- ルール
  win(X) :- move(X,Y) & NOT win(Y)
  EDB = { move(a,b) move(b,c) move(a,c) }
- M = EDB + {win(a), win(b)}
- r_1 : win(a) :- move(a,b) & NOT win(b)
  r_2 : win(b) :- move(b,c) & NOT win(c)
  r_3 : win(a) :- move(a,c) & NOT win(c)
- r_2, r_3 から win(a), win(b) が推論される.
- M は Stable Model

Gelfond-Lifschitz 変換

1. 任意のルールについて、すべての基礎代入例を生成.
2. 本体に偽となる EDB ファクトを含む基礎代入例を除く.
3. GL(M) = EDB + 本体が M で真となる頭部 (IDB ファクト) の全体.
4. GL(M) = M のとき M を Stable Model と呼ぶ.

- 極小 Stable Model が唯一つ存在するとき Stable 意味論が定義できるという.
- 但し、複数の極小 Stable Model の集合が意味をもつ場合もある.
  (後の講義で紹介)

例　デフォールト推論

- flies(X) :- bird(X) & NOT abnormal(X)
  abnormal(X) :- ostrich(X)
  bird(X) :- ostrich(X)
  bird(X) :- swallow(X)
  EDB = { ostrich(john), swallow(tom)}
- flies(john) :- bird(john) & NOT abnormal(john)　　flies(john) は推論できない
  flies(tom) :- bird(tom) & NOT abnormal(tom)　　flies(tom) を推論可能
- 極小 Stable Model
  M = EDB + { bird(john), bird(tom), abnormal(john), flies(tom)}
- 極小 Stable Model が唯一つ存在

例　フレーム問題

- loaded(0)

  loaded(T1) :- loaded(T0) & NOT shoots(T0) & succ(T0, T1)

  noise(T) :- loaded(T) & shoots(T)

  EDB = {shoots(2), succ(0,1), succ(1,2), ...}

- loaded(1) :- loaded(0) & NOT shoots(0) & succ(0, 1)

  loaded(2) :- loaded(1) & NOT shoots(1) & succ(1, 2)

  loaded(3) :- loaded(2) & NOT shoots(2) & succ(2, 3)

  $\vdots$

  noise(2) :- loaded(2) & shoots(2)

- 極小 Stable Model

  M = EDB + {loaded(0), loaded(1), loaded(2), noise(2)}

  NOT shoots(2) が偽, loaded(3) は推論されない.

- 極小 Stable Model が唯一つ存在

定理　(局所的) 層状論理プログラムにおいては, (局所的) 層状エルブランモデルが唯一の極小 Stable Model.
(注意: Stable Model は層の概念を使わなくても, 局所的層状論理プログラムに意味論を与えることに成功している.)

参考文献

- 局所的層状エルブランモデルを提案. Przymusinski, T.: On the declarative semantics of stratified deductive and logic programs. *Foundations of Deductive Databases and Logic Programming*, Morgan Kaufmann, 1988, pages 193-216.

- Gelfond-Lifschitz 変換の提案. Gelfond, M. and Lifschitz, V.: The stable model semantics for logic programming. *Proc. of the 5th Int'l Conf. and Symp. on Logic Programming*, IEEE, 1988, pages 1070-1080.

(局所的) 層状でない論理プログラムの極小 Stable Model

複数の極小 Stable Model が存在する場合

すなわち Stable 意味論が定義できない場合

- win(X) :- move(X,Y) & NOT win(Y)

  EDB = { move(a,b) move(b,a) }

- win(a) :- move(a,b) & NOT win(b)

  win(b) :- move(b,a) & NOT win(a)

  局所的層状論理プログラムにならないことに注意

- EDB + {win(a)}

  EDB + {win(b)}

  両方とも極小 Stable Model

唯一の極小 Stable Model が存在する場合（Stable 意味論が定義可能）

- p :- NOT q

  q :- NOT p

  p :- NOT p

- M = {p}

- Stable Model が局所的層状に比べて, より広いクラスのプログラムに意味を与えることができることを示唆している.

コメント

- 局所的層状や Stable Model の概念は 1986-8 年に提案される.

- 1988 年 Well-Founded Model の概念が提示され, より広範囲なプログラムに対して自然な意味が与えられるようになる.

問題

- p(X) :- arc(X,Y) & NOT q(Y)
  q(X) :- arc(X,Y) & NOT p(X)
  EDB = { arc(1,2), arc(2,3), arc(3,1) }
- 局所的層状論理プログラムか否か調べよ.
- Stable Model をすべて求めよ.
- 局所的層状論理プログラムとなる EDB の例を示せ.

問題

- p(X) :- a(X,Y) & q(Y) & NOT p(X)
  q(X) :- b(X,Y) & NOT q(Y)
  q(X) :- c(X)
  EDB = { a(1,2), a(2,3), a(3,1), b(2,1), b(3,2), c(1) }
- 局所的層状論理プログラムか否か調べよ.
- Stable Model をすべて求めよ.

問題

- a :- NOT b
  b :- NOT a
  p :- NOT p
- Stable Model をすべて求めよ.